

Introduction à GMSH

Nicolas KIELBASIEWICZ

5 mars 2010

GMSH est un logiciel libre créé par Christophe Geuzaine¹ et Jean-François Remacle², permettant de générer des maillages 2D et 3D de type éléments finis avec des outils de pré/post-traitement. C'est un logiciel multi-plateforme (Windows, Linux et Mac OS X) téléchargeable sur :

<http://geuz.org/gmsh>

L'objectif ici est de présenter les commandes élémentaires permettant de générer un maillage 2D étape par étape. Pour des informations détaillées sur les fonctionnalités de GMSH, je vous renvoie à la documentation disponible sur le site susnommé ou, pour simplement une introduction plus détaillée, sur ma page web : <http://www.ensta.fr/~kielbasi>.

Table des matières

1	Lancement du programme depuis une console	2
2	Définition des objets géométriques	3
2.1	Définition du rectangle	4
2.1.1	Définition des sommets	4
2.1.2	Définition des arêtes	5
2.2	Définition de l'ellipse	7
2.2.1	Définition des points de construction	7
2.2.2	Définition des arcs elliptiques	7
2.3	Définition de la surface plane	8
2.4	Définition des entités physiques	9
3	Génération du maillage	10
3.1	Le module <code>mesh</code>	10
3.2	Le format <code>msh</code>	10
3.2.1	L'entête	10

1. Professeur à l'Université de Liège : <http://www.montefiore.ulg.ac.be/~geuzaine>

2. Professeur à l'Université Catholique de Louvain <http://perso.uclouvain.be/jean-francois.remacle/>

3.2.2	Définition des nœuds	11
3.2.3	Définition des éléments	11
4	Le script .geo	12
4.1	Le script généré	12
4.1.1	Définition du rectangle	12
4.1.2	Définition de l'ellipse	13
4.1.3	Définition de la surface	13
4.1.4	Définition des labels	13
4.2	Une version directe	14

1 Lancement du programme depuis une console

Sous Windows et Mac OS, GMSH est une application native. Sous Linux, il se lance en ligne de commande, avec quelques options. Ce qu'il faut savoir au préalable, c'est que GMSH est un outil qui peut être utilisé directement en mode graphique ou purement en ligne de commande. En effet, en mode graphique, la génération des différents objets s'accompagne de la génération d'un fichier texte d'extension .geo qui contient la définition de l'ensemble des objets géométriques. Par ailleurs, comme son nom l'indique, GMSH est utilisé pour générer des maillages au format msh.

Voici quelques exemples de combinaisons d'options :

gmsh -help cette commande affiche l'aide avec la liste complète des options

gmsh -g file.geo cette commande démarre le script *file.geo* avec le module *geometry* (comportement par défaut)

gmsh -m file.geo cette commande démarre le script *file.geo* avec le module *mesh*

gmsh -s file.geo cette commande démarre le script *file.geo* avec le module *solver*

gmsh -p file.geo cette commande démarre le script *file.geo* avec le module *post-processing*

gmsh -1 file.geo -o maillage.msh cette commande génère le fichier *maillage.msh* à partir d'un maillage 1D sur la géométrie contenue dans *file.geo* et sort (mode console)

gmsh -2 -order 3 file.geo -o maillage.msh cette commande génère le fichier *maillage.msh* à partir d'un maillage 2D d'ordre 3 sur la géométrie contenue dans *file.geo* et sort (mode console)

gmsh -3 -clscale 0.1 file.geo -o maillage.msh cette commande génère le fichier *maillage.msh* à partir d'un maillage 3D de longueur caractéristique (=pas) 0.1 sur la géométrie contenue dans *file.geo* et sort (mode console)

Ces exemples font intervenir un certain nombre d'éléments que je vais expliquer à travers l'exemple d'un maillage rectangulaire possédant un trou elliptique.

GMSH est un logiciel possédant une interface graphique minimaliste, mais assez fonctionnelle. Elle se compose de deux frames : un menu, articulé en 4 modules, et une fenêtre principale (figure 1). Les boutons du menu donnent accès à des sous-menus.




FIGURE 1 – Exemple de maillage rectangulaire avec un trou elliptique.

2 Définition des objets géométriques

Le module `geometry` (figure 2) est disponible sur le menu par la liste de sélection ou le raccourci `g`.




FIGURE 2 – Menu du module `geometry`.

Il est dédié à la construction géométrique du maillage, avec la distinction entre les éléments purement géométrique utiles à la construction, et les éléments physiques du maillage à générer (avec leurs références, ...). C'est aussi par ce menu que l'on peut accéder au fichier de script (extension .geo) servant à générer le maillage, ainsi qu'à le recharger après modification directe. Je parlerai plus en détails de la syntaxe des instructions présentes dans ces fichiers dans la section 4.

2.1 Définition du rectangle

2.1.1 Définition des sommets

On veut définir les 4 sommets, respectivement aux coordonnées $(0, 0, 0)$, $(2, 0, 0)$, $(2, 1, 0)$ et $(0, 1, 0)$.




FIGURE 3 – Aspect du menu à chaque étape conduisant à la création d'un point par ouverture de la fenêtre contextuelle dédiée.

Pour cela, (figure 3) :

1. Lancer GMSH sur un fichier au nom de votre choix (par exemple *test.geo*).
2. Cliquer sur **Elementary entities**.
3. Cliquer sur **Add**.
4. Cliquer sur **New**.
5. Cliquer sur **Point** (figure 3).

Hormis les coordonnées spatiales du point, un champ très important à remplir est celui de la longueur caractéristique. Sa valeur représente la valeur du pas de maillage en ce point.

On va choisir la valeur 0.05.

On a ensuite la possibilité de positionner le point à la souris et d'appuyer sur la touche **e** pour ajouter un point, ou alors de remplir les 3 champs de coordonnées et de cliquer ensuite sur le bouton **Add**. On appuie sur **q** pour sortir du mode insertion (figure 4).




FIGURE 4 – Génération des 4 sommets du bord rectangulaire.

2.1.2 Définition des arêtes

À partir des points, on peut maintenant tracer les arêtes en reliant les sommets 2 à 2 :

1. Cliquer, depuis le menu `geometry`, sur `Elementary entities`.
2. Cliquer sur `Add`.
3. Cliquer sur `New`.
4. Cliquer sur `Straight Line`.
5. Cliquer sur le sommet de départ.
6. Cliquer sur le sommet d'arrivée.
7. Construire les 3 autres arêtes de la même manière.
8. Cliquer sur `q` pour quitter le mode insertion (figure 5).




FIGURE 5 – Génération des 4 arêtes du bord rectangulaire.

2.2 Définition de l'ellipse

Pour générer une ellipse, il faut générer 4 arcs elliptiques.

2.2.1 Définition des points de construction

Pour construire les arcs elliptiques, on a besoin des points de départ et d'arrivée de chaque arc, d'un point sur le grand axe de l'ellipse, et du centre de l'ellipse. Si l'on choisit comme points de constructions les 4 « sommets », alors il y a 5 points à construire.

1. Générer les points de coordonnées $(1, 0.5, 0)$, $(1.5, 0.5, 0)$, $(1, 0.8, 0)$, $(0.5, 0.5, 0)$, $(1, 0.2, 0)$ avec pour longueur caractéristique 0.05 (figure 6).




FIGURE 6 – Génération des 5 points de construction de l'ellipse.

2.2.2 Définition des arcs elliptiques

1. Cliquer, depuis le menu `geometry`, sur `Elementary entities`.
2. Cliquer sur `Add`.
3. Cliquer sur `New`.
4. Cliquer sur `Ellipse arc`.
5. Cliquer sur le point de droite.

6. Cliquer sur le centre de l'ellipse.
7. Cliquer sur le point de droite (il est sur le grand axe).
8. Cliquer sur le point du haut. Vous avez construit votre premier arc.
9. Construire les 3 autres arcs de la même manière.
10. Cliquer sur **q** pour quitter le mode insertion (figure 7).




FIGURE 7 – Génération des 4 arcs elliptiques.

2.3 Définition de la surface plane

Maintenant que nous avons défini les éléments des contours, on va définir la surface du domaine.

1. Cliquer, depuis le menu **geometry**, sur **Elementary entities**.
2. Cliquer sur **Add**.
3. Cliquer sur **New**.
4. Cliquer sur **Plane surface**.
5. Cliquer sur le contour principal (le rectangle).
6. Cliquer sur le trou (l'ellipse).
7. Cliquer sur **q** pour terminer la selection. Des traits en pointillé apparaissent (figure 8).




FIGURE 8 – Génération de la surface à mailler.

2.4 Définition des entités physiques

On va maintenant labelliser les différents éléments du maillages.

1. Cliquer, depuis le menu `geometry`, sur `Physical groups`.
2. Cliquer sur `Add`.
3. Cliquer sur `Line`.
4. Cliquer sur chacune des 4 arêtes du rectangle.
5. Cliquer sur `e`.
6. Cliquer sur chacun des 4 arcs de l'ellipse.
7. Cliquer sur `e`.
8. Cliquer sur `q` pour quitter.
9. Cliquer dans le menu sur `Surface`.
10. Cliquer sur l'un des traits pointillés.
11. Cliquer sur `e`.
12. Cliquer sur `q` pour quitter.

3 Génération du maillage

3.1 Le module mesh

Nous avons maintenant généré notre domaine géométrique. On peut maintenant générer le maillage, en sélectionnant le module mesh, soit grâce à la liste de sélection dans le menu, soit avec le raccourci m (figure 9).

1. Cliquer, dans le menu `mesh`, sur le bouton 2D.
2. Cliquer sur le bouton `Save`. Un fichier `test.msh` est créé (si le fichier de script est `test.geo`).




FIGURE 9 – Menu du module mesh.

3.2 Le format msh

3.2.1 L'entête

```
$MeshFormat  
2.2 0 8  
$EndMeshFormat
```

L'entête, représenté par un bloc `MeshFormat`, spécifie la version du format `msh` (2.2), que le maillage est un fichier texte (0) et la précision des flottants (8).

3.2.2 Définition des nœuds

```
$Nodes
732
1 0 0 0
2 2 0 0
3 2 1 0
4 0 1 0
5 1.5 0.5 0
6 1 0.8 0
7 0.5 0.5 0
8 1 0.2 0
9 0.05128205128192818 0 0
10 0.1025641025638564 0 0
```

Le bloc Nodes permet de spécifier le nombre de nœuds ainsi que les coordonnées de chaque nœud. Dans le maillage généré, il y a ici 732 nœuds, et le nœud 10 a pour coordonnées $(x, y, z) = (0.10256410 \dots, 0, 0)$.

3.2.3 Définition des éléments

```
$Elements
1464
1 1 2 12 1 1 9
2 1 2 12 1 9 10
3 1 2 12 1 10 11
4 1 2 12 1 11 12
```

Le bloc Elements permet de définir le nombre et l'ensemble des éléments du maillage (lignes, triangles, ...) à partir des références des nœuds qu'ils contiennent. Dans notre exemple, il y a donc 1464 éléments.

- La première colonne spécifie le numéro de l'élément.
- La deuxième colonne spécifie le type de l'élément (1 pour segment, 2 pour triangle, ...). Nous avons donc dans cet extrait du fichier uniquement des segments.
- La troisième colonne spécifie le nombre d'étiquettes associé à cet élément (ici, 2), correspondant aux colonnes suivantes. La première étiquette vaut ici 12, ce qui signifie que les segments présentés sont présents dans l'objet géométrique numéro 12. La seconde étiquette vaut ici 1, ce qui signifie que les segments sont sur le domaine physique numéroté 1, c'est à dire le rectangle.
- Les colonnes suivantes spécifient la liste des nœuds que l'élément contient. L'élément 1 est le segment reliant les nœuds 1 et 9, ...

Qu'en est-il des triangles ?

```
165 2 2 14 11 268 200 195
166 2 2 14 11 270 199 196
167 2 2 14 11 194 198 271
168 2 2 14 11 281 104 105
169 2 2 14 11 191 281 343
170 2 2 14 11 187 180 345
```

On retrouve la même structure. L'élément 165 est un triangle (2) ayant 2 étiquettes : l'élément géométrique qui le contient (14) et l'élément physique qui le contient (11), à savoir la surface. Logique!!! Ce triangle a pour sommets les nœuds 268, 200 et 195.

4 Le script .geo

4.1 Le script généré

GMSH génère en arrière plan un fichier de script d'extension .geo contenant la définition de tous les objets géométriques. Nous allons maintenant regarder la syntaxe des instructions présentes.

4.1.1 Définition du rectangle

```
Point (1) = {0, 0, 0, 0.05};
Point (2) = {2, 0, 0, 0.05};
Point (3) = {2, 1, 0, 0.05};
Point (4) = {0, 1, 0, 0.05};
Line (1) = {1, 2};
Line (2) = {2, 3};
Line (3) = {3, 4};
Line (4) = {4, 1};
```

On retrouve la définition des 4 sommets avec la commande **Point**, à partir de leurs coordonnées et de leur longueur caractéristique, entre accolades. Le nombre entre parenthèses désigne la référence géométrique du nœud. C'est cette référence qui est utilisée pour définir les segments avec la commande **Line**.

Les segments ont eux-même leurs références géométriques définies entre parenthèses et ils sont orientés, le premier point est le point de départ, le second est le point d'arrivée.

Toute instruction de définition / affectation se termine par un « ; ».

4.1.2 Définition de l'ellipse

```
Point (5) = {1, 0.5, 0, 0.05};  
Point (6) = {1.5, 0.5, 0, 0.05};  
Point (7) = {1, 0.8, 0, 0.05};  
Point (8) = {0.5, 0.5, 0, 0.05};  
Point (9) = {1, 0.2, 0, 0.05};  
Ellipse (5) = {6, 5, 6, 7};  
Ellipse (6) = {7, 5, 6, 8};  
Ellipse (7) = {8, 5, 6, 9};  
Ellipse (8) = {9, 5, 6, 6};
```

Les arcs elliptiques sont définis à l'aide de la commande **Ellipse** et les références des nœuds utilisés : dans l'ordre, on trouve le point de départ, le centre de l'ellipse, un point sur le grand axe, et le point d'arrivée.

4.1.3 Définition de la surface

```
Line Loop (9) = {3, 4, 1, 2};  
Line Loop (10) = {6, 7, 8, 5};  
Plane Surface (11) = {9, 10};
```

Pour définir une surface, il faut définir au préalable les contours sur lesquels elle s'appuie. Les contours sont définis à l'aide de la commande **Line Loop**, qui prend l'ensemble des lignes formant une ligne fermée. À noter que l'ordre dans lequel on a cliqué sur les 4 cotés du rectangle n'est pas nécessairement respecté ici. En effet, pour rédiger ce document, j'ai cliqué sur les cotés 3, 2, 1 et 4 dans cet ordre. Pour respecter cet ordre, il aurait fallu ajouter un signe « - » spécifiant que la ligne est parcourue dans le sens opposé à son orientation.

La surface, plane, est ici définie à l'aide de la commande **Plane Surface** à partir des 2 contours sur lesquels elle s'appuie.

4.1.4 Définition des labels

```
Physical Line (12) = {3, 2, 1, 4};  
Physical Line (13) = {5, 8, 7, 6};  
Physical Surface (14) = {11};
```

Le nombre entre parenthèses est la référence physique appliquée à l'ensemble des références géométriques entre accolades.

4.2 Une version directe

Si l'on choisit d'éditer le script directement (avec votre éditeur de texte favori), la seule règle fondamentale à respecter, hormis la syntaxe vue jusqu'ici, est la suivante : 2 objets de même dimension (points, lignes, surfaces, volumes) ne peuvent avoir la même référence géométrique. Cette dernière peut valoir n'importe quel entier naturel non nul.

L'exemple suivant montre comment vous auriez pu rédiger ce script en ajoutant quelques commentaires et en définissant la longueur caractéristique dans une variable.

```
lc=0.05;

// generation du rectangle
Point (1)={0,0,0,lc };
Point (2)={2,0,0,lc };
Point (3)={2,1,0,lc };
Point (4)={0,1,0,lc };

Line (1)={1,2};
Line (2)={2,3};
Line (3)={3,4};
Line (4)={4,1};

// generation de l'ellipse
Point (5)={1,0.5,0,lc };
Point (6)={1.5,0.5,0,lc };
Point (7)={1,0.75,0,lc };
Point (8)={0.5,0.5,0,lc };
Point (9)={1,0.25,0,lc };

Ellipse (5)={6,5,6,7};
Ellipse (6)={7,5,8,8};
Ellipse (7)={8,5,8,9};
Ellipse (8)={9,5,6,6};

// generation des contours
Line Loop(1)={1,2,3,4};
Line Loop(2)={5,6,7,8};

// generation de la surface
Plane Surface(1)={1,2};
```